9/8/99

PROBLEM SET ONE SOLUTIONS

Problem 1.

From the notes to lecture 1, the axial stress is given approximately by

```
STUDENT > sigma1:=-Mb/Iyy*y;
```

$$\sigma1 := -\frac{Mb\, y}{Iyy}$$

the shear stress by

```
STUDENT > tau:=V/2/Iyy*((h/2)^2-y^2);
```

$$\tau := \frac{1}{2}\,\frac{V\left(\frac{1}{4}h^2 - y^2\right)}{Iyy}$$

and the transverse normal stress by

```
STUDENT > sigma2:=1/6/Iyy*q*(3*(h/2)^2*y-y^3+2*(h/2)^3);
```

$$\sigma2 := \frac{1}{6}\,\frac{q\left(\frac{3}{4}h^2\, y - y^3 + \frac{1}{4}h^3\right)}{Iyy}$$

So that once we find q, V and Mb, we can calculate the (approximate) stresses from these formulae.

The beam is rectangular, say w by h, so that

```
STUDENT > Iyy:=1/12*w*h^3;
```

$$Iyy := \frac{1}{12}\, w\, h^3$$

and the loading is constant, so that

```
STUDENT > q:=P;
```

$$q := P$$

For **case (a)**, there will be a moment M1 at the left hand end, and reaction forces R1 and R2 at each end. The sum of the two reactions must equal PL. we can write the shear as

```
STUDENT > V:=-R1+q*x;
          M:=M1-int(V,x);
```

$$V := -R1 + P\, x$$

$$M := M1 + R1\, x - \frac{1}{2}P\, x^2$$

We obtain the transverse displacement by integrating the moment twice. The constants of integration are both zero because both the displacement and its derivative vanish at x = 0

```
STUDENT > int(M,x):EIyyv:=int(",x);
```

$$EIyyv := \frac{1}{2} M1\, x^2 + \frac{1}{6} R1\, x^3 - \frac{1}{24} P\, x^4$$

where E denotes Young's modulus.

M1 and R1 remain to be found,and they can be found from the condition that M and v vanish at x = L

```
STUDENT > x:=L:eqn1:=M;eqn2:=EIyyv;x:='x':
```

$$eqn1 := M1 + R1\, L - \frac{1}{2} P\, L^2$$

$$eqn2 := \frac{1}{2} M1\, L^2 + \frac{1}{6} R1\, L^3 - \frac{1}{24} P\, L^4$$

```
STUDENT > solve({eqn1,eqn2},{R1,M1});
```

$$\{R1 = \frac{5}{8} P\, L,\ M1 = -\frac{1}{8} P\, L^2\}$$

```
STUDENT > assign("):R2:=P*L-R1;
```

$$R2 := \frac{3}{8} P\, L$$

Now we can write out the shear, moment and deflection

```
STUDENT > shear:=factor(V);moment:=factor(M);deflection:=factor(EIyy
          v/E/Iyy);
```

$$shear := -\frac{1}{8} P\, (5\, L - 8\, x)$$

$$moment := -\frac{1}{8} P\, (L - x)\, (L - 4\, x)$$

$$deflection := -\frac{1}{4} \frac{P\, x^2\, (3\, L - 2\, x)\, (L - x)}{E\, w\, h^3}$$

and the approximate stresses are given by

```
STUDENT > axial_stress:=factor(subs(Mb=M,sigma1));
          shear_stress:=factor(tau);
          transverse_stress:=factor(sigma2);
```

$$axial\_stress := \frac{3}{2} \frac{P\, (L - x)\, (L - 4\, x)\, y}{w\, h^3}$$

$$shear\_stress := -\frac{3}{16} \frac{P\, (5\, L - 8\, x)\, (h - 2\, y)\, (h + 2\, y)}{w\, h^3}$$

$$transverse\_stress := \frac{1}{2} \frac{P\, (h - y)\, (h + 2\, y)^2}{w\, h^3}$$

For **case (b)** there is no moment at the left end, R1 = R2 = PL/2 by symmetry

For **case (a)**, there will be a moment M1 at the left hand end, and reaction forces R1 and R2 at each end. The sum of the two reactions must equal PL. we can write the shear as

```
STUDENT > R1:=P*L/2:R2:=P*L/2;
```

$$R2 := \frac{1}{2} P\, L$$

The first two integrations work the same way as in part (a)

```
STUDENT > V:=-R1+q*x;
          M:=M1-int(V,x);
```

$$V := -\frac{1}{2} P\, L + P\, x$$

$$M := \frac{1}{2} P\, L\, x - \frac{1}{2} P\, x^2$$

We obtain the transverse displacement by integrating the moment twice. The first constant of integration is undetermined because an initial slope is allowed for this case. The second is zeor because there is no displacment at x = 0

```
STUDENT > int(M,x)+c1:EIyyv:=int(",x);
```

$$EIyyv := \frac{1}{12} P\, L\, x^3 - \frac{1}{24} P\, x^4 + c1\, x$$

where E denotes Young's modulus.

c1 may be found from the condition of zero displacement at x = L

```
STUDENT > x:=L:eqn2:=EIyyv;x:='x':
```

$$eqn2 := \frac{1}{24} P\, L^4 + c1\, L$$

```
STUDENT > c1:=solve(eqn2,c1);
```

$$c1 := -\frac{1}{24} P\, L^3$$

Now we can write out the shear, moment and deflection

```
STUDENT > shear:=factor(V);moment:=factor(M);deflection:=factor(EIyy
          v/E/Iyy);
```

$$shear := -\frac{1}{2} P\, (L - 2\, x)$$

$$moment := \frac{1}{2} P\, x\, (L - x)$$

$$deflection := -\frac{1}{2} \frac{P\, x\, (L - x)\, (-x^2 + L\, x + L^2)}{E\, w\, h^3}$$

and the approximate stresses are given by

```
STUDENT > axial_stress:=factor(subs(Mb=M,sigma1));
          shear_stress:=factor(tau);
          transverse_stress:=factor(sigma2);
```

$$axial\_stress := -6 \frac{P x (L - x) y}{w h^3}$$

$$shear\_stress := -\frac{3}{4} \frac{P (L - 2x)(h - 2y)(h + 2y)}{w h^3}$$

$$transverse\_stress := \frac{1}{2} \frac{P (h - y)(h + 2y)^2}{w h^3}$$

Finally, for case (c) there will be a moment at each end and a reaction at each end. These will be equal to each other by symmetry, so we will have

```
STUDENT > R1:=P*L/2:R2:=P*L/2:M1:='M1':M2:=M1;
```

$$M2 := M1$$

The first two integrations work the same way as in part (a)

```
STUDENT > V:=-R1+q*x;
          M:=M1-int(V,x);
```

$$V := -\frac{1}{2} P L + P x$$

$$M := M1 + \frac{1}{2} P L x - \frac{1}{2} P x^2$$

We obtain the transverse displacement by integrating the moment twice. As in part (a) both constants of integration are zero because the slope and displacement vanish at $x = 0$

```
STUDENT > int(M,x):EIyyv:=int(",x);
```

$$EIyyv := \frac{1}{2} M1\, x^2 + \frac{1}{12} P L x^3 - \frac{1}{24} P x^4$$

where E denotes Young's modulus.

M1 may be found from the condition of zero displacement at x = L

```
STUDENT > x:=L:eqn2:=EIyyv;x:='x':
```

$$eqn2 := \frac{1}{2} M1\, L^2 + \frac{1}{24} P L^4$$

```
STUDENT > M1:=solve(eqn2,M1);
```

$$M1 := -\frac{1}{12} P L^2$$

Now we can write out the shear, moment and deflection

```
STUDENT > shear:=factor(V);moment:=factor(M);deflection:=factor(EIyy
          v/E/Iyy);
```

$$shear := -\frac{1}{2} P (L - 2x)$$

$$moment := -\frac{1}{12} P (L^2 - 6 L x + 6 x^2)$$

$$deflection := -\frac{1}{2}\frac{P\,x^2\,(L-x)^2}{E\,w\,h^3}$$

and the approximate stresses are given by

```
STUDENT > axial_stress:=factor(subs(Mb=M,sigma1));
          shear_stress:=factor(tau);
          transverse_stress:=factor(sigma2);
```

$$axial\_stress := \frac{P\,(L^2-6\,L\,x+6\,x^2)\,y}{w\,h^3}$$

$$shear\_stress := -\frac{3}{4}\frac{P\,(L-2\,x)\,(h-2\,y)\,(h+2\,y)}{w\,h^3}$$

$$transverse\_stress := \frac{1}{2}\frac{P\,(h-y)\,(h+2\,y)^2}{w\,h^3}$$

Some comments are required here. We never imposed the condition of zero slope at x = L, but it came out correctly. This is because of the symmetry of the problem. If we had not assumed symmetry, we would have had extra conditions to determine R1 and R2 and M1 and M2 separately, and we would have been able to calculate the result we assumed.

```
STUDENT > save `ps1.m`:
```

Problem 2.

This has proved to be more interestng than I had expected. I will have a detailed exploration of this at a later date. Stay tuned. For now . . .

Let sigma1 denote the radial normal stress, sigma2 the tangential (theta) normal stress and tau denote the shear (r-theta) stress. Define F as the integral of sigma2 from R1 to R2, V as the integral from r = R1 to R2 of tau and M the integral of (r - (R1 + R2)/2) times sigma2 from r = R1 to R2. These play the same role as F, V and M in the flat beam model. They will be functions of theta in general, and it can be shown without too much difficulty that the following equations must be zero

```
STUDENT > F:='F':V:='V':M:='M':
          f1:=diff(F(theta),theta)+V(theta);
          f2:=diff(V(theta),theta)-F(theta)+p(theta)*R1;
          m1:=diff(M(theta),theta)+V(theta)*(R1+R2)/2;
```

$$f1 := \left(\frac{\partial}{\partial\theta}\mathrm{F}(\theta)\right)+\mathrm{V}(\theta)$$

$$f2 := \left(\frac{\partial}{\partial\theta}\mathrm{V}(\theta)\right)-\mathrm{F}(\theta)+\mathrm{p}(\theta)\,R1$$

$$m1 := \left(\frac{\partial}{\partial\theta}\mathrm{M}(\theta)\right)+\frac{1}{2}\mathrm{V}(\theta)\,(R1+R2)$$

For the problem as posed p is a constant, but the general problem would have it vary along the curved beam.

If the beam is thin, the R1 ≈ R2 and we can write

```
STUDENT > R1:=R-h/2:R2:=R+h/2:f1;f2;m1;
```

$$\left(\frac{\partial}{\partial\theta}\mathrm{F}(\theta)\right)+\mathrm{V}(\theta)$$
$$\left(\frac{\partial}{\partial\theta}\mathrm{V}(\theta)\right)-\mathrm{F}(\theta)+\mathrm{p}(\theta)\left(R-\frac{1}{2}h\right)$$
$$\left(\frac{\partial}{\partial\theta}\mathrm{M}(\theta)\right)+\mathrm{V}(\theta)\,R$$

To look at the limit of large radius of curvature, approximated by R, let theta = x/R and redefine the functions as functions of x

```
STUDENT > f2:=subs(p(theta)=px(x),f2):
          f1x:=subsop(1=R*diff(Fx(x),x),2=Vx(x),f1);
          f2x:=subsop(1=R*diff(Vx(x),x),2=Fx(x),f2);
          m1x:=subsop(1=R*diff(Mx(x),x),2=R*Vx(x),m1);
```

$$f1x := R\left(\frac{\partial}{\partial x}\mathrm{Fx}(x)\right)+\mathrm{Vx}(x)$$
$$f2x := R\left(\frac{\partial}{\partial x}\mathrm{Vx}(x)\right)+\mathrm{Fx}(x)+\mathrm{px}(x)\left(R-\frac{1}{2}h\right)$$
$$m1x := R\left(\frac{\partial}{\partial x}\mathrm{Mx}(x)\right)+R\,\mathrm{Vx}(x)$$

and in the limit that R —> ∞, you can see that the equations reduce to the beam equations. I can do this by writing

```
STUDENT > R:=1/epsilon:
          factor(epsilon*f1x):epsilon:=0:"";
          epsilon:='epsilon':factor(epsilon*f2x):epsilon:=0:"";
          epsilon:='epsilon':factor(epsilon*m1x):epsilon:=0:"";
          epsilon:='epsilon':
```

$$\frac{\partial}{\partial x}\mathrm{Fx}(x)$$
$$\left(\frac{\partial}{\partial x}\mathrm{Vx}(x)\right)+\mathrm{px}(x)$$
$$\left(\frac{\partial}{\partial x}\mathrm{Mx}(x)\right)+\mathrm{Vx}(x)$$

and these (equated to zero) are clearly recognizable as the straight beam equations.

Dealing with displacements is messier than you might expect at first sight, and I'll get to that another time.